
=====================Simple_Clocked_ADC===================================== 

1) Adding a sample and hold to a clockless ADC will convert it to a clocked ADC.
2) The sampling process appears to generate a spectrum which modulates the input
   signal around the sampling frequency. 
3) This more visible when the sampled signal is subtracted from the input. 
4) The vin signal generate harmonics above and below mulitples of the sample frequency.

Sampling at 8msec is a 125Hz sample rate. A 3Hz input signal will produce harmonics
at 125Hz +/-3Hz, and 250Hz +/-3Hz, and 373Hz +/-3Hz  etc...

===================MacSpiceCode====================================================
Simple_Clocked_ADC
*
*       VIN
*      ________________    __________    __________    __________    __________    __________    __________    
*    _|_     | TH0    _|_  | TH1    _|_  | TH0    _|_  | TH1    _|_  | TH0    _|_  | TH1    _|_  | TH0    _|_    TH1    
*   /_  \    |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \ |      _/   \       
*  // \  \   | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/ | |\  | \___/   |\  
*  \   \//   |_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |___|_| \_|   |_____| \_
*   \___/      | / | OFF0    | / |  OFF1   | / | OFF2    | / |  OFF3   | / | OFF4    | / |  OFF5   | / | OFF6    | / |  
*    _|_       |/  |         |/  |         |/  |         |/  |         |/  |         |/  |         |/  |         |/  |
*    ///          _|_           _|_           _|_           _|_           _|_           _|_           _|_           _|_
*                |D0 |         |D1 |         |D2 |         |D3 |         |D4 |         |D5 |         |D6 |         |D7 |
*                |___|         |___|         |___|         |___|         |___|         |___|         |___|         |___|
*
*        ________
*        __|     \
*        __|      \
*        __| DAC   \__ VOUTA
*        __|       /
*        __|      /
*        __|_____/
*
*         
*   VOUTA     IN2                                  
*    ___  |\    /         IN3 |\      ___
*   |IN |_|1\__/  ____________|2\____|OUT|
*   |___| | / S1   _|_        | /    |___|
*         |/       ___        |/           
*                C1 |               
*                  _|_                
*                  /// 

*=========Create_Signal==================
*V_SIN#   NODE_P NODE_N DC     VALUE  SIN(   V_DC   AC_MAG FREQ   DELAY  FDamp)
VIN       VIN    0      DC     0      SIN(   .5      .5     3                  )
*V_PULSE# NODE_P NODE_N DC     VALUE  PULSE( VINIT  VPULSE TDELAY TRISE  TFALL  PWIDTH PERIOD )
VCLKL     CLK    0      DC     0      PULSE( 0      1      1n     1n     1n     4m     8m     )

XPOSE1    CLK    CNTL   POS_E 
XS_H1     VOUTA  CNTL   VOUT      SH 



BTH0      D0     0      V =    u( V(VIN)  -1/2)
BOFF0     VIN    OFF0   V =    V(D0)/2
BTH1      D1     0      V =    u( V(OFF0) -1/4)
BOFF1     OFF0   OFF1   V =    V(D1)/4
BTH2      D2     0      V =    u( V(OFF1) -1/8)
BOFF2     OFF1   OFF2   V =    V(D2)/8
BTH3      D3     0      V =    u( V(OFF2) -1/16)
BOFF3     OFF2   OFF3   V =    V(D3)/16
BTH4      D4     0      V =    u( V(OFF3) -1/32)
BOFF4     OFF3   OFF4   V =    V(D4)/32
BTH5      D5     0      V =    u( V(OFF4) -1/64)
BOFF5     OFF4   OFF5   V =    V(D5)/64
BTH6      D6     0      V =    u( V(OFF5) -1/128)
BOFF6     OFF5   OFF6   V =    V(D6)/128
BTH7      D7     0      V =    u( V(OFF6) -1/256)
BOFF7     OFF6   OFF7   V =    V(D7)/245
BDAC      VOUTA  0      V =    V(D0)/2+V(D1)/4+V(D2)/8+V(D3)/16+V(D4)/32+V(D5)/64+V(D6)/128+V(D7)/245

.control
*TRAN     TSTEP  TSTOP  TSTART TMAX   ?UIC?
tran      .05m    1      0      .05m   
set       pensize = 2
plot      vin  vouta vout
plot      vin -vout xlimit 1m 1
echo             "===================FFT_and_Plot====================================="
linearize 
let              FFT_BandWidth_Hz =  1k
let              FFT_resolution_Hz = 1
echo             "FFT_BandWidth_Hz=  $&FFT_BandWidth_Hz"
echo             "FFT_resolution_Hz= $&FFT_resolution_Hz"
set              specwindow =         "rectangular"
spec             $&FFT_resolution_Hz  $&FFT_BandWidth_Hz   $&FFT_resolution_Hz     v(vout)
plot             mag (vout)   loglog
echo             "===================Done============================================="

.endc

*=========Sample_Hold============================================
*         
*            IN2                                  
*    ___  |\    /         IN3 |\      ___
*   |IN |_|1\__/  ____________|2\____|OUT|
*   |___| | / S1   _|_        | /    |___|
*         |/       ___        |/           
*                C1 |               
*                  _|_                
*                  /// 
.SUBCKT   SH     IN     CNTL      OUT   
 B1       IN2    0      V =       v(IN )
 S1       IN2    IN3    CNTL     0    SW 
 C1       IN3    0      .1u
 R1       IN3    0      100Meg
 B2       OUT    0      V =       v(IN3 )
.ENDS     SH

*=========POS_Edge============================================
*
*  ___   |\             VBF ____                ___
* |IN |__|3\________________|   \   ___    ___  |  \   ___
* |___|  | / |           VLP| PE \_|OUT|  |IN |_|PE \_|OUT|
*        |/  |/\  /\  /\____|    / |___|  |___| |   / |___|
*               \/  \/   |  |___/               |__/
*               RLP     _|_
*                       ___
*                   CLP  |
*                       _|_
*                       ///
*
.SUBCKT   POS_E  IN     OUT   
 BBUF     VBF    0      V =   u( v(IN )-.5 )
 RLP      VBF    VLP    10k
 CLP      VLP    0      1n   IC=0
 BAND     OUT    0      V =    u( u(v(VBF )-.5)*u(.5 -v(VLP ) ) -.1)
.ENDS     POS_E

.MODEL    SW      SW(    VT=.5 VH=.1  RON=1 ROFF=100MEG)

.end

